# NAG Library Routine Document

# F07VEF (DTBTRS)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

## 1 Purpose

F07VEF (DTBTRS) solves a real triangular band system of linear equations with multiple right-hand sides, $AX = B$ or $A^{\mathrm{T}}X = B$.

## 2 Specification

```
SUBROUTINE F07VEF(UPLO, TRANS, DIAG, N, KD, NRHS, AB, LDAB, B, LDB,
1                 INFO)
INTEGER           N, KD, NRHS, LDAB, LDB, INFO
double precision  AB(LDAB,*), B(LDB,*)
CHARACTER*1       UPLO, TRANS, DIAG
```

The routine may be called by its LAPACK name ***dtbtrs***.

## 3 Description

F07VEF (DTBTRS) solves a real triangular band system of linear equations $AX = B$ or $A^{\mathrm{T}}X = B$.

## 4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

Higham N J (1989) The accuracy of solutions to triangular systems *SIAM J. Numer. Anal.* **26** 1252–1265

## 5 Parameters

1: UPLO – CHARACTER*1 *Input*

*On entry*: indicates whether $A$ is upper or lower triangular.

UPLO = 'U'
$A$ is upper triangular.

UPLO = 'L'
$A$ is lower triangular.

*Constraint*: UPLO = 'U' or 'L'.

2: TRANS – CHARACTER*1 *Input*

*On entry*: indicates the form of the equations.

TRANS = 'N'
The equations are of the form $AX = B$.

TRANS = 'T' or 'C'
The equations are of the form $A^{\mathrm{T}}X = B$.

*Constraint*: TRANS = 'N', 'T' or 'C'.

3: DIAG – CHARACTER*1 *Input*

*On entry*: indicates whether $A$ is a nonunit or unit triangular matrix.

DIAG = 'N'
$A$ is a nonunit triangular matrix.

DIAG = 'U'
$A$ is a unit triangular matrix; the diagonal elements are not referenced and are assumed to be 1.

*Constraint*: DIAG = 'N' or 'U'.

4: N – INTEGER *Input*

*On entry*: $n$, the order of the matrix $A$.

*Constraint*: N $\geq 0$.

5: KD – INTEGER *Input*

*On entry*: $k_d$, the number of superdiagonals of the matrix $A$ if UPLO = 'U', or the number of subdiagonals if UPLO = 'L'.

*Constraint*: KD $\geq 0$.

6: NRHS – INTEGER *Input*

*On entry*: $r$, the number of right-hand sides.

*Constraint*: NRHS $\geq 0$.

7: AB(LDAB,$*$) – ***double precision*** array *Input*

**Note**: the second dimension of the array AB must be at least $\max(1, \text{N})$.

*On entry*: the $n$ by $n$ triangular band matrix $A$.

The matrix is stored in rows 1 to $k_d + 1$, more precisely,

if UPLO = 'U', the elements of the upper triangle of $A$ within the band must be stored with element $A_{ij}$ in $\text{AB}(k_d + 1 + i - j, j)$ for $\max(1, j - k_d) \leq i \leq j$;

if UPLO = 'L', the elements of the lower triangle of $A$ within the band must be stored with element $A_{ij}$ in $\text{AB}(1 + i - j, j)$ for $j \leq i \leq \min(n, j + k_d)$.

If DIAG = 'U', the diagonal elements of $A$ are assumed to be 1, and are not referenced.

8: LDAB – INTEGER *Input*

*On entry*: the first dimension of the array AB as declared in the (sub)program from which F07VEF (DTBTRS) is called.

*Constraint*: LDAB $\geq$ KD + 1.

9: B(LDB,$*$) – ***double precision*** array *Input/Output*

**Note**: the second dimension of the array B must be at least $\max(1, \text{NRHS})$.

*On entry*: the $n$ by $r$ right-hand side matrix $B$.

*On exit*: the $n$ by $r$ solution matrix $X$.

10: LDB – INTEGER *Input*

*On entry*: the first dimension of the array B as declared in the (sub)program from which F07VEF (DTBTRS) is called.

*Constraint*: LDB $\geq \max(1, \text{N})$.

11: INFO – INTEGER *Output*

*On exit*: INFO $= 0$ unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO $< 0$

If INFO $= -i$, the $i$th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO $> 0$

If INFO $= i$, $a(i,i)$ is exactly zero; $A$ is singular and the solution has not been computed.

## 7 Accuracy

The solutions of triangular systems of equations are usually computed to high accuracy. See Higham (1989).

For each right-hand side vector $b$, the computed solution $x$ is the exact solution of a perturbed system of equations $(A+E)x = b$, where

$$|E| \leq c(k)\epsilon|A|,$$

$c(k)$ is a modest linear function of $k$, and $\epsilon$ is the ***machine precision***.

If $\hat{x}$ is the true solution, then the computed solution $x$ satisfies a forward error bound of the form

$$\frac{\|x-\hat{x}\|_\infty}{\|x\|_\infty} \leq c(k)\,\mathrm{cond}(A,x)\epsilon, \qquad \text{provided} \qquad c(k)\,\mathrm{cond}(A,x)\epsilon < 1,$$

where $\mathrm{cond}(A,x) = \left\| |A^{-1}||A||x| \right\|_\infty / \|x\|_\infty$.

Note that $\mathrm{cond}(A,x) \leq \mathrm{cond}(A) = \left\| |A^{-1}||A| \right\|_\infty \leq \kappa_\infty(A)$; $\mathrm{cond}(A,x)$ can be much smaller than $\mathrm{cond}(A)$ and it is also possible for $\mathrm{cond}\left(A^{\mathrm{T}}\right)$ to be much larger (or smaller) than $\mathrm{cond}(A)$.

Forward and backward error bounds can be computed by calling F07VHF (DTBRFS), and an estimate for $\kappa_\infty(A)$ can be obtained by calling F07VGF (DTBCON) with NORM = 'I'.

## 8 Further Comments

The total number of floating-point operations is approximately $2nkr$ if $k \ll n$.

The complex analogue of this routine is F07VSF (ZTBTRS).

## 9 Example

This example solves the system of equations $AX = B$, where

$$A = \begin{pmatrix} -4.16 & 0.00 & 0.00 & 0.00 \\ -2.25 & 4.78 & 0.00 & 0.00 \\ 0.00 & 5.86 & 6.32 & 0.00 \\ 0.00 & 0.00 & -4.82 & 0.16 \end{pmatrix} \quad \text{and} \quad B = \begin{pmatrix} -16.64 & -4.16 \\ -13.78 & -16.59 \\ 13.10 & -4.94 \\ -14.14 & -9.96 \end{pmatrix}.$$

Here $A$ is treated as a lower triangular band matrix with one subdiagonal.

## 9.1 Program Text

```
*     F07VEF Example Program Text
*     Mark 15 Release. NAG Copyright 1991.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, KDMAX, LDAB, NRHMAX, LDB
      PARAMETER        (NMAX=8,KDMAX=NMAX,LDAB=KDMAX+1,NRHMAX=NMAX,
     +                 LDB=NMAX)
      CHARACTER        TRANS, DIAG
      PARAMETER        (TRANS='N',DIAG='N')
*     .. Local Scalars ..
      INTEGER          I, IFAIL, INFO, J, KD, N, NRHS
      CHARACTER        UPLO
*     .. Local Arrays ..
      DOUBLE PRECISION AB(LDAB,NMAX), B(LDB,NRHMAX)
*     .. External Subroutines ..
      EXTERNAL         DTBTRS, X04CAF
*     .. Intrinsic Functions ..
      INTRINSIC        MAX, MIN
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F07VEF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N, KD, NRHS
      IF (N.LE.NMAX .AND. KD.LE.KDMAX .AND. NRHS.LE.NRHMAX) THEN
*
*        Read A and B from data file
*
         READ (NIN,*) UPLO
         IF (UPLO.EQ.'U') THEN
            DO 20 I = 1, N
               READ (NIN,*) (AB(KD+1+I-J,J),J=I,MIN(N,I+KD))
   20       CONTINUE
         ELSE IF (UPLO.EQ.'L') THEN
            DO 40 I = 1, N
               READ (NIN,*) (AB(1+I-J,J),J=MAX(1,I-KD),I)
   40       CONTINUE
         END IF
         READ (NIN,*) ((B(I,J),J=1,NRHS),I=1,N)
*
*        Compute solution
*
         CALL DTBTRS(UPLO,TRANS,DIAG,N,KD,NRHS,AB,LDAB,B,LDB,INFO)
*
*        Print solution
*
         WRITE (NOUT,*)
         IF (INFO.EQ.0) THEN
            IFAIL = 0
            CALL X04CAF('General',' ',N,NRHS,B,LDB,'Solution(s)',IFAIL)
         ELSE
            WRITE (NOUT,*) 'A is singular'
         END IF
      END IF
*
      END
```

## 9.2 Program Data

```
F07VEF Example Program Data
  4  1  2                    :Values of N, KD and NRHS
  'L'                        :Value of UPLO
 -4.16
 -2.25   4.78
         5.86   6.32
```

```
               -4.82   0.16   :End of matrix A
-16.64  -4.16
-13.78 -16.59
 13.10  -4.94
-14.14  -9.96                 :End of matrix B
```

## 9.3 Program Results

```
 F07VEF Example Program Results

 Solution(s)
              1          2
 1      4.0000     1.0000
 2     -1.0000    -3.0000
 3      3.0000     2.0000
 4      2.0000    -2.0000
```